Pensieve Header: Computing ρ_2 efficiently.

```
SetDirectory["~/AcademicPensieve/Talks/KnotTheoryCongress-2502/Rho2Data"];
<< "../../../Projects/KnotTheory/KnotTheory/init.m"
AppendTo[$Path, KnotTheoryDirectory[] = "../../../Projects/KnotTheory"];
Print[$Path = Cases[$Path, _String]];
<< "../Rot.m"
```

```
Loading KnotTheory` version of October 29, 2024, 10:29:52.1301.
Read more at http://katlas.org/wiki/KnotTheory.
Loading Rot.m from http://drorbn.net/ktc25/ap to compute rotation numbers.
```

In[∘]:=
```
T2z[p_] := Module[{q = Expand[p], n, c},
   If[q === 0, 0, c = Coefficient[q, T, n = Exponent[q, T]];
    c z^(2 n) + T2z[q - c (T^(1/2) - T^(-1/2))^(2 n)]]];
```

Pre-computing the Feynman Diagrams

In[∘]:=
```
CF[ℰ_] := Expand@Collect[ℰ, g__, F] /. F → Factor;
```

In[∘]:=
```
{p*, x*, π*, ξ*} = {π, ξ, p, x};  (u_i_)* := (u*)_i;
```

In[∘]:=
```
Zip_{}[ℰ_] := ℰ;
Zip_{ξ_, ξs___}[ℰ_] := (Collect[ℰ // Zip_{ξs}, ξ] /. f_. ξ^d_. :→ (D[f, {ξ*, d}])) /. ξ* → 0
```

In[∘]:=
```
px2g[ℰ_] := CF@Module[{ps, xs, Q, α, β},
   ps = Union[Cases[ℰ, p__, ∞]]; xs = Union[Cases[ℰ, x__, ∞]];
   Q = Sum[p0* x0* g_{p0[[2]], x0[[2]]}, {p0, ps}, {x0, xs}];
   Expand[Zip_{ps⋃xs}[ℰ ⅇ^Q]]
  ]
```

In[∘]:= $\text{px2g}\left[\frac{s}{2}\left(x_i\,(p_i-p_j)\left(\left(T^s-1\right)x_i\,p_j+2\,(1-p_j\,x_j)\right)-1\right)\right]$

Out[∘]=

$$-\frac{s}{2}+s\,g_{i,i}-s\,g_{j,i}+s\,(-1+T^s)\,g_{i,i}\,g_{j,i}-s\,g_{i,j}\,g_{j,i}-s\,(-1+T^s)\,g_{j,i}^2-s\,g_{i,i}\,g_{j,j}+2\,s\,g_{j,i}\,g_{j,j}$$

From AcademicPensieve/Talks/Beijing-2407/More.nb:

In[∘]:=

$$q[s\_, i\_, j\_] := x_i\,(p_{i+1} - p_i) + x_j\,(p_{j+1} - p_j) + (T^s - 1)\,x_i\,(p_{i+1} - p_{j+1});$$

$$r_1[s\_, i\_, j\_] := \frac{s}{2}\,\left(x_i\,(p_i - p_j)\,\left((T^s - 1)\,x_i\,p_j + 2\,(1 - p_j\,x_j)\right) - 1\right);$$

$$\begin{aligned} r_2[1, i\_, j\_] :=\ & \Big(-6\,p_i\,x_i + 6\,p_j\,x_i - 3\,(-1 + 3\,T)\,p_i\,p_j\,x_i^2 + 3\,(-1 + 3\,T)\,p_j^2\,x_i^2 + 4\,(-1 + T)\,p_i^2\,p_j\,x_i^3 - \\ & 2\,(-1 + T)\,(5 + T)\,p_i\,p_j^2\,x_i^3 + 2\,(-1 + T)\,(3 + T)\,p_j^3\,x_i^3 + 18\,p_i\,p_j\,x_i\,x_j - 18\,p_j^2\,x_i\,x_j - \\ & 6\,p_i^2\,p_j\,x_i^2\,x_j + 6\,(2 + T)\,p_i\,p_j^2\,x_i^2\,x_j - 6\,(1 + T)\,p_j^3\,x_i^2\,x_j - 6\,p_i\,p_j^2\,x_i\,x_j^2 + 6\,p_j^3\,x_i\,x_j^2\Big) \Big/ 12; \end{aligned}$$

$$\begin{aligned} r_2[-1, i\_, j\_] :=\ & \Big(-6\,T^2\,p_i\,x_i + 6\,T^2\,p_j\,x_i + 3\,(-3 + T)\,T\,p_i\,p_j\,x_i^2 - 3\,(-3 + T)\,T\,p_j^2\,x_i^2 - 4\,(-1 + T)\,T\,p_i^2\,p_j\,x_i^3 + \\ & 2\,(-1 + T)\,(1 + 5\,T)\,p_i\,p_j^2\,x_i^3 - 2\,(-1 + T)\,(1 + 3\,T)\,p_j^3\,x_i^3 + 18\,T^2\,p_i\,p_j\,x_i\,x_j - \\ & 18\,T^2\,p_j^2\,x_i\,x_j - 6\,T^2\,p_i^2\,p_j\,x_i^2\,x_j + 6\,T\,(1 + 2\,T)\,p_i\,p_j^2\,x_i^2\,x_j - \\ & 6\,T\,(1 + T)\,p_j^3\,x_i^2\,x_j - 6\,T^2\,p_i\,p_j^2\,x_i\,x_j^2 + 6\,T^2\,p_j^3\,x_i\,x_j^2\Big) \Big/ (12\,T^2); \end{aligned}$$

$$\gamma_1[\varphi\_, k\_] := \varphi\,(1/2 - x_k\,p_k);$$

$$\gamma_2[\varphi\_, k\_] := -\varphi^2\,p_k\,x_k/2;$$

$$\mathcal{L}[X_{i\_,j\_}[s\_]] := T^{s/2}\,\mathbb{E}\left[q[s, i, j] + \epsilon\,r_1[s, i, j] + \epsilon^2\,r_2[s, i, j] + O[\epsilon]^3\right];$$

$$\mathcal{L}[C_{k\_}[\varphi\_]] := T^{\varphi/2}\,\mathbb{E}\left[-x_k\,(p_k - p_{k+1}) + \epsilon\,\gamma_1[\varphi, k] + \epsilon^2\,\gamma_2[\varphi, k] + O[\epsilon]^3\right];$$

$$\mathcal{L}[K\_] := (2\,\pi)^{-\text{Features}[K][[1]]}\,\text{CF}[\mathcal{L}\,/@\,\text{Features}[K][[2]]];$$

```
vs[K_] := Union @@ Table[{p_i, x_i}, {i, Features[K][[1]]}]
```

In[∘]:=

$$F[X_{i\_,j\_}[1]] = \text{px2g}\left[r_2[1, i, j] + \frac{r_1[1, i, j]^2}{2}\right]$$

Out[∘]=

$$\begin{aligned} & \frac{1}{8} - g_{i,i} + g_{i,i}^2 + g_{j,i} + (-1 - 2\,T)\,g_{i,i}\,g_{j,i} + 5\,(-1 + T)\,g_{i,i}^2\,g_{j,i} + 2\,g_{i,j}\,g_{j,i} - 6\,g_{i,i}\,g_{i,j}\,g_{j,i} + 2\,T\,g_{j,i}^2 - \\ & (-1 + T)\,(11 + T)\,g_{i,i}\,g_{j,i}^2 + 3\,(-1 + T)^2\,g_{i,i}^2\,g_{j,i}^2 + (6 + T)\,g_{i,j}\,g_{j,i}^2 - 6\,(-1 + T)\,g_{i,i}\,g_{i,j}\,g_{j,i}^2 + \\ & 2\,g_{i,j}^2\,g_{j,i}^2 + (-1 + T)\,(6 + T)\,g_{j,i}^3 - 6\,(-1 + T)^2\,g_{i,i}\,g_{j,i}^3 + 6\,(-1 + T)\,g_{i,j}\,g_{j,i}^3 + 3\,(-1 + T)^2\,g_{j,i}^4 + \\ & 2\,g_{i,i}\,g_{j,j} - 3\,g_{i,i}^2\,g_{j,j} - 4\,g_{j,i}\,g_{j,j} + 2\,(6 + T)\,g_{i,i}\,g_{j,i}\,g_{j,j} - 6\,(-1 + T)\,g_{i,i}^2\,g_{j,i}\,g_{j,j} - \\ & 2\,g_{i,j}\,g_{j,i}\,g_{j,j} + 8\,g_{i,i}\,g_{i,j}\,g_{j,i}\,g_{j,j} - 3\,(3 + T)\,g_{j,i}^2\,g_{j,j} + 18\,(-1 + T)\,g_{i,i}\,g_{j,i}^2\,g_{j,j} - 12\,g_{i,j}\,g_{j,i}^2\,g_{j,j} - \\ & 12\,(-1 + T)\,g_{j,i}^3\,g_{j,j} - g_{i,i}\,g_{j,j}^2 + 2\,g_{i,i}^2\,g_{j,j}^2 + 3\,g_{j,i}\,g_{j,j}^2 - 12\,g_{i,i}\,g_{j,i}\,g_{j,j}^2 + 12\,g_{j,i}^2\,g_{j,j}^2 \end{aligned}$$

In[∘]:=
```
F[X_{i_,j_}[-1]] = px2g[r_2[-1, i, j] + r_1[-1, i, j]^2/2]
```
Out[∘]=

$$\frac{1}{8}-g_{i,i}+g_{i,i}^2+g_{j,i}-\frac{(2+T)\,g_{i,i}\,g_{j,i}}{T}-\frac{5\,(-1+T)\,g_{i,i}^2\,g_{j,i}}{T}+2\,g_{i,j}\,g_{j,i}-6\,g_{i,i}\,g_{i,j}\,g_{j,i}+\frac{2\,g_{j,i}^2}{T}+\frac{(-1+T)\,(1+11\,T)\,g_{i,i}\,g_{j,i}^2}{T^2}+\frac{3\,(-1+T)^2\,g_{i,i}^2\,g_{j,i}^2}{T^2}+\frac{(1+6\,T)\,g_{i,j}\,g_{j,i}^2}{T}+\frac{6\,(-1+T)\,g_{i,i}\,g_{i,j}\,g_{j,i}^2}{T}+2\,g_{i,j}^2\,g_{j,i}^2-\frac{(-1+T)\,(1+6\,T)\,g_{j,i}^3}{T^2}-\frac{6\,(-1+T)^2\,g_{i,i}\,g_{j,i}^3}{T^2}-\frac{6\,(-1+T)\,g_{i,j}\,g_{j,i}^3}{T}+\frac{3\,(-1+T)^2\,g_{j,i}^4}{T^2}+2\,g_{i,i}\,g_{j,j}-3\,g_{i,i}^2\,g_{j,j}-4\,g_{j,i}\,g_{j,j}+\frac{2\,(1+6\,T)\,g_{i,i}\,g_{j,i}\,g_{j,j}}{T}+\frac{6\,(-1+T)\,g_{i,i}^2\,g_{j,i}\,g_{j,j}}{T}-2\,g_{i,j}\,g_{j,i}\,g_{j,j}+8\,g_{i,i}\,g_{i,j}\,g_{j,i}\,g_{j,j}-\frac{3\,(1+3\,T)\,g_{j,i}^2\,g_{j,j}}{T}-\frac{18\,(-1+T)\,g_{i,i}\,g_{j,i}^2\,g_{j,j}}{T}-12\,g_{i,j}\,g_{j,i}^2\,g_{j,j}+\frac{12\,(-1+T)\,g_{j,i}^3\,g_{j,j}}{T}-g_{i,i}\,g_{j,j}^2+2\,g_{i,i}^2\,g_{j,j}^2+3\,g_{j,i}\,g_{j,j}^2-12\,g_{i,i}\,g_{j,i}\,g_{j,j}^2+12\,g_{j,i}^2\,g_{j,j}^2$$

In[∘]:=
```
Short[F[X_{i0_,j0_}[1] X_{i1_,j1_}[1]] = px2g[r_1[1, i0, j0] r_1[1, i1, j1]]]
```
Out[∘]//Short=

$$\frac{1}{4}-\frac{g_{i0,i0}}{2}+\ll 274\gg+4\,g_{j0,i0}\,g_{j0,i1}\,g_{j1,j0}\,g_{j1,j1}$$

In[∘]:=
```
Short[F[X_{i0_,j0_}[1] X_{i1_,j1_}[-1]] = px2g[r_1[1, i0, j0] r_1[-1, i1, j1]]]
```
Out[∘]//Short=

$$-\frac{1}{4}+\frac{g_{i0,i0}}{2}-g_{i0,i1}\,g_{i1,i0}+\ll 302\gg+2\,g_{i0,i0}\,g_{j0,i1}\,g_{j1,j0}\,g_{j1,j1}-4\,g_{j0,i0}\,g_{j0,i1}\,g_{j1,j0}\,g_{j1,j1}$$

In[∘]:=
```
Short[F[X_{i0_,j0_}[-1] X_{i1_,j1_}[-1]] = px2g[r_1[-1, i0, j0] r_1[-1, i1, j1]]]
```
Out[∘]//Short=

$$\frac{1}{4}-\frac{g_{i0,i0}}{2}+\ll 304\gg+4\,g_{j0,i0}\,g_{j0,i1}\,g_{j1,j0}\,g_{j1,j1}$$

In[∘]:=
```
F[C_{k_}[1]] = px2g[γ_2[1, k] + γ_1[1, k]^2/2]
```
Out[∘]=

$$\frac{1}{8}-g_{k,k}+g_{k,k}^2$$

In[∘]:=
```
F[C_{k_}[-1]] = px2g[γ_2[-1, k] + γ_1[-1, k]^2/2]
```
Out[∘]=

$$\frac{1}{8}-g_{k,k}+g_{k,k}^2$$

In[∘]:=
```
F[X_{i_,j_}[1] C_{k_}[1]] = px2g[r_1[1, i, j] γ_1[1, k]]
```

Out[∘]=

$$-\frac{1}{4}+\frac{g_{i,i}}{2}-\frac{g_{j,i}}{2}+\frac{1}{2}(-1+T)\,g_{i,i}\,g_{j,i}-\frac{1}{2}g_{i,j}\,g_{j,i}+\frac{1}{2}(1-T)\,g_{j,i}^2-\frac{1}{2}g_{i,i}\,g_{j,j}+g_{j,i}\,g_{j,j}-g_{i,k}\,g_{k,i}+(1-T)\,g_{i,k}\,g_{j,i}\,g_{k,i}+g_{i,k}\,g_{j,j}\,g_{k,i}+g_{j,k}\,g_{k,i}+(1-T)\,g_{i,i}\,g_{j,k}\,g_{k,i}+g_{i,j}\,g_{j,k}\,g_{k,i}+2(-1+T)\,g_{j,i}\,g_{j,k}\,g_{k,i}-2\,g_{j,j}\,g_{j,k}\,g_{k,i}+g_{i,k}\,g_{j,i}\,g_{k,j}+g_{i,i}\,g_{j,k}\,g_{k,j}-2\,g_{j,i}\,g_{j,k}\,g_{k,j}+\frac{g_{k,k}}{2}-g_{i,i}\,g_{k,k}+g_{j,i}\,g_{k,k}+(1-T)\,g_{i,i}\,g_{j,i}\,g_{k,k}+g_{i,j}\,g_{j,i}\,g_{k,k}+(-1+T)\,g_{j,i}^2\,g_{k,k}+g_{i,i}\,g_{j,j}\,g_{k,k}-2\,g_{j,i}\,g_{j,j}\,g_{k,k}$$

In[∘]:=
```
F[X_{i_,j_}[1] C_{k_}[-1]] = px2g[r_1[1, i, j] γ_1[-1, k]]
```

Out[∘]=

$$\frac{1}{4}-\frac{g_{i,i}}{2}+\frac{g_{j,i}}{2}+\frac{1}{2}(1-T)\,g_{i,i}\,g_{j,i}+\frac{1}{2}g_{i,j}\,g_{j,i}+\frac{1}{2}(-1+T)\,g_{j,i}^2+\frac{1}{2}g_{i,i}\,g_{j,j}-g_{j,i}\,g_{j,j}+g_{i,k}\,g_{k,i}+(-1+T)\,g_{i,k}\,g_{j,i}\,g_{k,i}-g_{i,k}\,g_{j,j}\,g_{k,i}-g_{j,k}\,g_{k,i}+(-1+T)\,g_{i,i}\,g_{j,k}\,g_{k,i}-g_{i,j}\,g_{j,k}\,g_{k,i}-2(-1+T)\,g_{j,i}\,g_{j,k}\,g_{k,i}+2\,g_{j,j}\,g_{j,k}\,g_{k,i}-g_{i,k}\,g_{j,i}\,g_{k,j}-g_{i,i}\,g_{j,k}\,g_{k,j}+2\,g_{j,i}\,g_{j,k}\,g_{k,j}-\frac{g_{k,k}}{2}+g_{i,i}\,g_{k,k}-g_{j,i}\,g_{k,k}+(-1+T)\,g_{i,i}\,g_{j,i}\,g_{k,k}-g_{i,j}\,g_{j,i}\,g_{k,k}+(1-T)\,g_{j,i}^2\,g_{k,k}-g_{i,i}\,g_{j,j}\,g_{k,k}+2\,g_{j,i}\,g_{j,j}\,g_{k,k}$$

In[∘]:=
```
F[X_{i_,j_}[-1] C_{k_}[1]] = px2g[r_1[-1, i, j] γ_1[1, k]]
```

Out[∘]=

$$\frac{1}{4}-\frac{g_{i,i}}{2}+\frac{g_{j,i}}{2}+\frac{(-1+T)\,g_{i,i}\,g_{j,i}}{2\,T}+\frac{1}{2}g_{i,j}\,g_{j,i}-\frac{(-1+T)\,g_{j,i}^2}{2\,T}+\frac{1}{2}g_{i,i}\,g_{j,j}-g_{j,i}\,g_{j,j}+g_{i,k}\,g_{k,i}-\frac{(-1+T)\,g_{i,k}\,g_{j,i}\,g_{k,i}}{T}-g_{i,k}\,g_{j,j}\,g_{k,i}-g_{j,k}\,g_{k,i}-\frac{(-1+T)\,g_{i,i}\,g_{j,k}\,g_{k,i}}{T}-g_{i,j}\,g_{j,k}\,g_{k,i}+\frac{2(-1+T)\,g_{j,i}\,g_{j,k}\,g_{k,i}}{T}+2\,g_{j,j}\,g_{j,k}\,g_{k,i}-g_{i,k}\,g_{j,i}\,g_{k,j}-g_{i,i}\,g_{j,k}\,g_{k,j}+2\,g_{j,i}\,g_{j,k}\,g_{k,j}-\frac{g_{k,k}}{2}+g_{i,i}\,g_{k,k}-g_{j,i}\,g_{k,k}-\frac{(-1+T)\,g_{i,i}\,g_{j,i}\,g_{k,k}}{T}-g_{i,j}\,g_{j,i}\,g_{k,k}+\frac{(-1+T)\,g_{j,i}^2\,g_{k,k}}{T}-g_{i,i}\,g_{j,j}\,g_{k,k}+2\,g_{j,i}\,g_{j,j}\,g_{k,k}$$

In[∘]:=
```
F[X_{i_,j_}[-1] C_{k_}[-1]] = px2g[r1[-1, i, j] γ1[-1, k]]
```

Out[∘]=

$$-\frac{1}{4}+\frac{g_{i,i}}{2}-\frac{g_{j,i}}{2}-\frac{(-1+T)\,g_{i,i}\,g_{j,i}}{2\,T}-\frac{1}{2}\,g_{i,j}\,g_{j,i}+\frac{(-1+T)\,g_{j,i}^2}{2\,T}-\frac{1}{2}\,g_{i,i}\,g_{j,j}+g_{j,i}\,g_{j,j}-g_{i,k}\,g_{k,i}+\frac{(-1+T)\,g_{i,k}\,g_{j,i}\,g_{k,i}}{T}+g_{i,k}\,g_{j,j}\,g_{k,i}+g_{j,k}\,g_{k,i}+\frac{(-1+T)\,g_{i,i}\,g_{j,k}\,g_{k,i}}{T}+g_{i,j}\,g_{j,k}\,g_{k,i}-\frac{2\,(-1+T)\,g_{j,i}\,g_{j,k}\,g_{k,i}}{T}-2\,g_{j,j}\,g_{j,k}\,g_{k,i}+g_{i,k}\,g_{j,i}\,g_{k,j}+g_{i,i}\,g_{j,k}\,g_{k,j}-2\,g_{j,i}\,g_{j,k}\,g_{k,j}+\frac{g_{k,k}}{2}-g_{i,i}\,g_{k,k}+g_{j,i}\,g_{k,k}+\frac{(-1+T)\,g_{i,i}\,g_{j,i}\,g_{k,k}}{T}+g_{i,j}\,g_{j,i}\,g_{k,k}-\frac{(-1+T)\,g_{j,i}^2\,g_{k,k}}{T}+g_{i,i}\,g_{j,j}\,g_{k,k}-2\,g_{j,i}\,g_{j,j}\,g_{k,k}$$

In[∘]:=
```
F[C_{k0_}[1] C_{k1_}[1]] = px2g[γ1[1, k0] γ1[1, k1]]
```

Out[∘]=

$$\frac{1}{4}-\frac{g_{k0,k0}}{2}+g_{k0,k1}\,g_{k1,k0}-\frac{g_{k1,k1}}{2}+g_{k0,k0}\,g_{k1,k1}$$

In[∘]:=
```
F[C_{k0_}[1] C_{k1_}[-1]] = px2g[γ1[1, k0] γ1[-1, k1]]
```

Out[∘]=

$$-\frac{1}{4}+\frac{g_{k0,k0}}{2}-g_{k0,k1}\,g_{k1,k0}+\frac{g_{k1,k1}}{2}-g_{k0,k0}\,g_{k1,k1}$$

In[∘]:=
```
F[C_{k0_}[-1] C_{k1_}[-1]] = px2g[γ1[-1, k0] γ1[-1, k1]]
```

Out[∘]=

$$\frac{1}{4}-\frac{g_{k0,k0}}{2}+g_{k0,k1}\,g_{k1,k0}-\frac{g_{k1,k1}}{2}+g_{k0,k0}\,g_{k1,k1}$$

A ρ_2 program

In[∘]:=
```
ρ2[K_] := Module[{n, Fs, Xs, Cs, A, w, φ, Δ, G},
  {n, Fs} = List @@ Features[K]; {++n, Xs = Cases[Fs, X__[_]], Cs = Cases[Fs, C_[_]]};
  A = IdentityMatrix[n]; w = φ = 0;
  Xs /. X_{i_,j_}[s_] :> (w += s; A[[{i, j}, {i + 1, j + 1}]] += ( -T^s  T^s - 1 ) );
                                                                 (  0     -1    )
  Cs /. C_{k_}[ϕ_] :> (φ += ϕ; A[[k, k + 1]] += -1);
  Δ = T^(-(φ+w)/2) Det[A];
  G = Inverse[A];
  Factor[2 Δ^4 Total@Factor[(F /@ Subsets[Fs, {1, 2}]) /. g_{α_,β_} :> G[[α, β]]]]
 ]
```

In[∘]:=
```
ρ2[Knot[3, 1]] // T2z
```
Out[∘]=

$2 - 4 z^2 + 3 z^4 + 4 z^6 + z^8$

In[∘]:=
```
DunfieldKnots = ReadList["../../People/Dunfield/nmd_random_knots"] /. k_Integer :> k + 1;
DK[n_] := DunfieldKnots[[n - 2]];
```

In[∘]:=
```
ρ2[DK[3]] // T2z
```
Out[∘]=

$2 - 4 z^2 + 3 z^4 + 4 z^6 + z^8$

In[∘]:=
```
AbsoluteTiming[ρ2[DK[10]] // T2z]
```
Out[∘]=

$\{0.72777, 18 - 188 z^2 + 235 z^4 + 27660 z^6 + 213594 z^8 + 821660 z^{10} + 1954273 z^{12} + 3121080 z^{14} + 3488879 z^{16} + 2791080 z^{18} + 1613408 z^{20} + 673212 z^{22} + 200257 z^{24} + 41304 z^{26} + 5600 z^{28} + 448 z^{30} + 16 z^{32}\}$

In[∘]:=
```
AbsoluteTiming[ρ2[DK[20]] // T2z]
```
Out[∘]=

$\{9.0306, 6 - 36 z^2 + 137 z^4 + 684 z^6 - 1791 z^8 - 5096 z^{10} + 1422 z^{12} + 17992 z^{14} + 27682 z^{16} - 18008 z^{18} - 57311 z^{20} - 37936 z^{22} + 21378 z^{24} + 70060 z^{26} + 44443 z^{28} - 22340 z^{30} - 42159 z^{32} - 15648 z^{34} + 5483 z^{36} + 7252 z^{38} + 3030 z^{40} + 672 z^{42} + 80 z^{44} + 4 z^{46}\}$

In[∘]:=
```
AbsoluteTiming[ρ2[DK[30]] // T2z]
```
Out[∘]=

$\{15.8729, 18 + 68 z^2 + 923 z^4 + 23140 z^6 + 171996 z^8 + 635240 z^{10} + 1540635 z^{12} + 2631544 z^{14} + 4066922 z^{16} + 9460936 z^{18} + 27345191 z^{20} + 64880584 z^{22} + 126266074 z^{24} + 236701404 z^{26} + 453939012 z^{28} + 798652060 z^{30} + 1157856095 z^{32} + 1321551392 z^{34} + 1168261597 z^{36} + 790184656 z^{38} + 401546393 z^{40} + 148778040 z^{42} + 38549286 z^{44} + 7212168 z^{46} + 1662954 z^{48} + 676208 z^{50} + 219256 z^{52} + 38752 z^{54} + 3360 z^{56}\}$

In[◦]:=
```
AbsoluteTiming[ρ₂[DK[100]] // T2z]
```

Out[◦]=

$$\{1291.56,\ 34 - 1488\,z^{2} - 39\,548\,z^{4} + 10\,792\,z^{6} + 6\,833\,710\,z^{8} + 37\,437\,252\,z^{10} -$$
$$658\,965\,945\,z^{12} - 10\,119\,271\,592\,z^{14} - 48\,325\,133\,574\,z^{16} + 45\,188\,512\,768\,z^{18} +$$
$$1\,623\,056\,339\,094\,z^{20} + 8\,042\,542\,914\,076\,z^{22} + 13\,862\,292\,743\,997\,z^{24} - 42\,367\,394\,004\,452\,z^{26} -$$
$$328\,639\,475\,841\,479\,z^{28} - 882\,404\,199\,820\,520\,z^{30} - 678\,734\,897\,597\,277\,z^{32} +$$
$$2\,968\,778\,845\,001\,816\,z^{34} + 13\,483\,687\,905\,184\,821\,z^{36} + 53\,424\,952\,939\,386\,272\,z^{38} +$$
$$261\,111\,433\,750\,380\,252\,z^{40} + 893\,796\,479\,152\,524\,888\,z^{42} + 908\,491\,620\,268\,861\,274\,z^{44} -$$
$$7\,381\,286\,401\,581\,790\,692\,z^{46} - 43\,829\,571\,230\,826\,922\,603\,z^{48} - 116\,615\,001\,093\,219\,977\,808\,z^{50} -$$
$$108\,551\,039\,981\,570\,119\,802\,z^{52} + 429\,396\,631\,623\,058\,448\,676\,z^{54} + 2\,300\,749\,537\,571\,497\,873\,155\,z^{56} +$$
$$5\,834\,088\,556\,718\,901\,798\,572\,z^{58} + 8\,742\,287\,978\,829\,460\,305\,042\,z^{60} + 3\,993\,498\,361\,982\,634\,813\,196\,z^{62} -$$
$$18\,080\,889\,302\,516\,101\,195\,176\,z^{64} - 58\,803\,981\,399\,441\,041\,942\,560\,z^{66} -$$
$$93\,703\,489\,310\,740\,333\,663\,047\,z^{68} - 64\,798\,608\,219\,905\,577\,791\,124\,z^{70} +$$
$$93\,311\,684\,654\,539\,439\,636\,594\,z^{72} + 386\,441\,395\,637\,408\,144\,772\,028\,z^{74} +$$
$$691\,311\,967\,013\,150\,282\,998\,277\,z^{76} + 764\,621\,852\,353\,528\,805\,859\,752\,z^{78} +$$
$$393\,981\,892\,304\,860\,346\,898\,846\,z^{80} - 356\,737\,777\,026\,879\,287\,459\,532\,z^{82} -$$
$$1\,002\,587\,760\,029\,621\,732\,082\,622\,z^{84} - 821\,992\,142\,044\,076\,482\,857\,404\,z^{86} +$$
$$610\,517\,509\,924\,259\,061\,381\,641\,z^{88} + 2\,897\,927\,772\,963\,644\,056\,170\,416\,z^{90} +$$
$$4\,844\,952\,255\,264\,945\,708\,133\,288\,z^{92} + 5\,239\,595\,014\,022\,424\,564\,638\,868\,z^{94} +$$
$$3\,864\,641\,961\,945\,495\,817\,201\,657\,z^{96} + 1\,789\,584\,147\,863\,217\,802\,183\,116\,z^{98} +$$
$$515\,312\,585\,443\,744\,008\,803\,160\,z^{100} + 680\,984\,153\,458\,413\,027\,019\,624\,z^{102} +$$
$$1\,587\,604\,611\,214\,351\,611\,193\,742\,z^{104} + 1\,999\,078\,542\,660\,880\,921\,427\,784\,z^{106} +$$
$$1\,345\,741\,423\,688\,618\,967\,472\,227\,z^{108} + 112\,054\,426\,675\,992\,909\,636\,980\,z^{110} -$$
$$809\,180\,314\,402\,316\,085\,439\,650\,z^{112} - 949\,860\,317\,832\,176\,961\,547\,548\,z^{114} -$$
$$497\,638\,085\,180\,674\,510\,602\,584\,z^{116} + 49\,719\,287\,470\,682\,048\,547\,672\,z^{118} +$$
$$315\,174\,300\,298\,616\,952\,503\,039\,z^{120} + 238\,365\,667\,911\,770\,956\,838\,224\,z^{122} +$$
$$11\,903\,825\,805\,629\,930\,779\,277\,z^{124} - 133\,908\,640\,060\,649\,997\,199\,828\,z^{126} -$$
$$120\,584\,724\,683\,583\,260\,100\,659\,z^{128} - 30\,573\,577\,986\,680\,147\,242\,056\,z^{130} +$$
$$29\,787\,342\,776\,252\,248\,291\,008\,z^{132} + 33\,316\,926\,602\,725\,257\,170\,992\,z^{134} +$$
$$13\,249\,578\,355\,492\,116\,209\,280\,z^{136} - 466\,389\,804\,957\,125\,331\,968\,z^{138} -$$
$$3\,474\,520\,182\,461\,126\,135\,009\,z^{140} - 2\,199\,495\,177\,129\,275\,098\,632\,z^{142} -$$
$$732\,062\,729\,882\,109\,656\,245\,z^{144} + 60\,908\,557\,973\,391\,345\,752\,z^{146} + 264\,443\,411\,436\,407\,131\,086\,z^{148} +$$
$$150\,905\,870\,551\,228\,610\,632\,z^{150} + 8\,263\,949\,831\,072\,721\,160\,z^{152} - 40\,077\,705\,419\,978\,461\,732\,z^{154} -$$
$$24\,796\,560\,868\,735\,382\,995\,z^{156} - 3\,354\,736\,603\,841\,509\,844\,z^{158} + 3\,789\,620\,594\,671\,053\,397\,z^{160} +$$
$$2\,344\,996\,074\,617\,217\,784\,z^{162} + 308\,247\,572\,187\,169\,218\,z^{164} - 191\,143\,606\,999\,749\,796\,z^{166} -$$
$$52\,763\,721\,834\,069\,815\,z^{168} + 27\,781\,158\,896\,637\,420\,z^{170} + 14\,714\,777\,332\,724\,951\,z^{172} -$$
$$520\,843\,538\,566\,648\,z^{174} - 1\,754\,643\,449\,107\,029\,z^{176} - 247\,179\,575\,500\,996\,z^{178} +$$
$$127\,730\,443\,998\,943\,z^{180} + 52\,708\,567\,383\,940\,z^{182} + 6\,213\,607\,654\,940\,z^{184} -$$
$$250\,636\,503\,912\,z^{186} - 83\,287\,401\,093\,z^{188} + 3\,970\,217\,700\,z^{190} + 1\,076\,058\,594\,z^{192}\}$$

The Run

In[◦]:=
```
Table[n → NumberOfKnots[n], {n, 3, 15}]
```

Out[◦]=

{3 → 1, 4 → 1, 5 → 2, 6 → 3, 7 → 7, 8 → 21, 9 → 49,
10 → 165, 11 → 552, 12 → 2176, 13 → 9988, 14 → 46 972, 15 → 253 293}

```
AbsoluteTiming[
 Table[Echo[K] → T2z[ρ₂[K]], {K, AllKnots[{15, 15}]}] >> "Rho2_15.m"
]
```

```
Quit[]
```